/*
Carl Palmer	
Proposal title: Personality and Resistance to Persuasion


HYPOTHESES

Stated-Hyp1: Personality affects how people process information involving risk... 
"for the risk frames, we would expect to see higher levels of support for risky policies across the board for the Extraverted and those Open to Experiences, while the risk averse tendency of those higher in Conscientiousness should, in these cases, offset their natural inclination to favor the policies"
"Low emotional stability... should decreases support for issues framed as risky" (p. 3)

	Test-Hyp1: Extraversion will increase opposition for iran strike in framed vs. stripped condition
	
	Test-Hyp2: Openness will increase opposition for iran strike in framed vs. stripped condition

	Test-Hyp3: Conscientiousness will decrease opposition for iran strike in framed vs. stripped condition		

	Test-Hyp4: Neuroticism will decrease opposition for iran strike in framed vs. stripped condition		
	
Stated-Hyp2: Personality affects how people process information involving altruistic giving. "For issues with altruistic frames, we would expect to see greater support among individuals higher in Agreeableness and Conscientiousness" (p. 3)

	Test-Hyp5: Agreeableness will increase opposition to oil drilling in framed vs. stripped condition
	
	Test-Hyp6: Conscientiousness will increase opposition to oil drilling in framed vs. stripped condition
		
	Test-Hyp7: Agreeableness will increase support for environmental regulations in framed vs. stripped condition
	
	Test-Hyp8: Conscientiousness will increase support for environmental regulations in framed vs. stripped condition
	
Stated-Hyp3: There is an interaction effect b/w personality traits and how individuals respond to persuasive appeals... (p. 3)
"we expect risk frames will have less of an effect on [Individuals low in Neuroticism and Extraversion]" (p. 4)
"those high in Neuroticism and Conscientiousness... [will] react more strongly to risk frames" (p.4)
"Altruistic frames should have a greater effect on [Those high in both Extraversion and Agreeableness]" (p.4)
"Altruistic frames should have a greater effect on [Those high in both Extraversion and Conscientiousness]" (p.4)

	Test-Hyp9: Jointly higher levels of neuroticism and extraversion will decrease the effect of the risk frame on support for Iran strike.
	
	Test-Hyp10: Jointly higher levels of neuroticism and conscientiousness will increase the effect of the risk frame on support for Iran strike.
	
	Test-Hyp11: Jointly higher levels of agreeableness and extraversion will increase the effect of the altruism frame on opposition to oil drilling.
	
	Test-Hyp12: Jointly higher levels of agreeableness and extraversion will increase the effect of the altruism frame on support for environmental regulations.
	
	Test-Hyp13: Jointly higher levels of conscientiousness and extraversion will increase the effect of the altruism frame on opposition to oil drilling.
	
	Test-Hyp14: Jointly higher levels of conscientiousness and extraversion will increase the effect of the altruism frame on support for environmental regulations.
		

Stated-Hyp4: "we also expect to see larger effects generally for those higher in Openness to Experience" (p. 4)

	Test-Hyp15: Openness will increase the effect of risk frame on opposition to Iran strike. 
	Test-Hyp16: Openness will increase the effect of altruism frame on opposition to oil drilling. 
	Test-Hyp17: Openness will increase the effect of altruism frame on support for environmental regulations. 
		
********************************************************************************
NOTES:
- personality trait measures are constructed based on the description of the traits in the proposal. 
- Low emotional stability in this context appears to mean the same thing as neuroticism.
- There are two risk frames in this study.
This is one of the frames: 
"	2c.	Unmanned drones help reduce casualties among US soldiers, while also potentially causing casualties among civilians.  Do you favor or oppose the use of unmanned drones by the US in combat zones?"
	
- Not sure if the riskier option is the one that causes more soldiers to die or the one that causes more civilians to die. 
*/

clear all
use "Palmer347.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* stripped or framed (risk and altruism)
	tab XTESS128
	recode XTESS128 (1=1 "framed") (2=0 "stripped"), gen(framed)
	
* OUTCOME MEASURES

* policy opposition questions
	foreach var in Q2a Q2b Q2c Q2d Q3a Q3b Q3c Q3d {
		replace `var'=. if `var'==-1
		tab `var', mis
	}

* risk frames 	
	/*
	a.	Advocates of a US invasion of Iran suggest a preemptive strike on Iranian nuclear facilities would make Americans safer, while critics argue that an invasion of Iran would put Americans at greater risk.  Do you favor or oppose US military intervention in Iran?

	*/
	
	gen iranstrike_opp = Q2a
	replace iranstrike_opp=Q3a if iranstrike_opp==.
	tab iranstrike_opp
	// higher values indicate higher risk preference
	
	
* altruistic frames (recode so higher values indicate more pro-sociality)

	/*
	b.	While expanding offshore drilling may increase damage done to the ecosystem, it has the potential to increase energy independence and reduce costs to consumers.  Do you favor or oppose increasing oil drilling off the US coast?	
		
	d.	Proponents of stricter environmental regulations argue that we need to preserve the environment for future generations.  Critics of stricter regulations suggest that efforts will strongly inhibit economic growth.  Do you favor or oppose stronger environmental regulations for businesses?
	*/	
 	
	gen oildrill_opp = Q2b 
	replace oildrill_opp=Q3b if oildrill_opp==.
	tab oildrill_opp
	// higher values indicate opposition of drilling
	
	gen envreg_support= 6-Q2d
	replace envreg_support=6-Q3d if envreg_support==.
	tab envreg_support
	// higher values indicate support
		
* MODERATORS

* Personality traits
	foreach var in Q1a Q1b Q1c Q1d Q1e Q1f Q1g Q1h Q1i Q1j {
		replace `var'=. if `var'==-1
		tab `var'
	}
	
	/*
	a.	Extraverted, enthusiastic
	b.	Critical, quarrelsome
	c.	Dependable, self-disciplined
	d.	Anxious, easily upset
	e.	Open to new experiences, complex
	f.	Reserved, quiet
	g.	Sympathetic, warm
	h.	Disorganized, careless
	i.	Calm, emotionally stable
	j.	Conventional, uncreative
	*/
	
	* extraversion
	clonevar ext=Q1a 
	gen int_inverse= 8 - Q1f
	gen extraversion = ext+ int_inverse
	tab extraversion
	
	* agreeableness
	gen agree1 = 8-Q1b
	gen agree2 = Q1g
	gen agreeableness= agree1+ agree2
	tab agreeableness
	
	* conscientiousness
	clonevar consc1= Q1c
	gen consc2= 8 - Q1h 	
	gen conscientiousness= consc1+ consc2
	tab conscientiousness
	
	* neuroticism
	clonevar neur1 =Q1d
	gen neur2 = 8-Q1i
	gen neuroticism = neur1 + neur2
	tab neuroticism
	
	* openness
	clonevar open1= Q1e 
	gen open2= 8-Q1j 
	gen openness= open1+open2
	tab openness
	
********************************************************************************

* ANALYSIS
	

*Test-Hyp1: Extraversion will increase opposition for iran strike (high risk).
	reg iranstrike_opp c.extraversion##i.framed
	// reject. 0.251
	tess c.extraversion#1.framed +, init(Palmer347) 
	
*Test-Hyp2: Openness will increase opposition for iran strike (high risk).	
	reg iranstrike_opp c.openness##i.framed
	// reject. 0.283
	tess c.openness#1.framed +
	
*Test-Hyp3: Conscientiousness will decrease opposition for iran strike (high risk).		
	reg iranstrike_opp c.conscientiousness##i.framed
	// do not reject. 0.961
	tess c.conscientiousness#1.framed -
	
*Test-Hyp4: Neuroticism will decrease opposition for iran strike (high risk).	
	reg iranstrike_opp c.neuroticism##i.framed
	// reject. 0.253
	tess c.neuroticism#1.framed -
	
*Test-Hyp5: Agreeableness will increase opposition to oil drilling.
	reg oildrill_opp c.agreeableness##i.framed
	// reject. 0.524
	tess c.agreeableness#1.framed +
	
*Test-Hyp6: Conscientiousness will increase opposition to oil drilling.
	reg oildrill_opp c.conscientiousness##i.framed
	// reject. 0.793
	tess c.conscientiousness#1.framed +
	
*Test-Hyp7: Agreeableness will increase support for environmental regulations.
	reg envreg_support c.agreeableness##i.framed
	// reject. 0.824
	tess c.agreeableness#1.framed +
	
*Test-Hyp8: Conscientiousness will increase support for environmental regulations.
	reg envreg_support c.conscientiousness##i.framed
	// reject. 0.586	
	tess c.conscientiousness#1.framed +
	
*Test-Hyp9: Jointly higher levels of neuroticism and extraversion will decrease the effect of the risk frame on support for Iran strike.
	reg iranstrike_opp c.neuroticism##c.extraversion##i.framed
	// reject. 0.705
	tess c.neuroticism#c.extraversion#1.framed -
	
*Test-Hyp10: Jointly higher levels of neuroticism and conscientiousness will increase the effect of the risk frame on support for Iran strike.
	reg iranstrike_opp c.neuroticism##c.conscientiousness##i.framed
	// reject. 0.972
	tess c.neuroticism#c.conscientiousness#1.framed +,
	
*Test-Hyp11: Jointly higher levels of agreeableness and extraversion will increase the effect of the altruism frame on opposition to oil drilling.
	reg oildrill_opp c.agreeableness##c.extraversion##i.framed
	// reject. 0.525	
	tess c.agreeableness#c.extraversion#1.framed +
	
*Test-Hyp12: Jointly higher levels of agreeableness and extraversion will increase the effect of the altruism frame on support for environmental regulations.
	reg envreg_support c.agreeableness##c.extraversion##i.framed
	// reject. 0.961	
	tess c.agreeableness#c.extraversion#1.framed +
	
*Test-Hyp13: Jointly higher levels of conscientiousness and extraversion will increase the effect of the altruism frame on opposition to oil drilling.
	reg oildrill_opp c.conscientiousness##c.extraversion##i.framed
	// reject. 0.718	
	tess c.conscientiousness#c.extraversion#1.framed +
	
*Test-Hyp14: Jointly higher levels of conscientiousness and extraversion will increase the effect of the altruism frame on support for environmental regulations.
	reg envreg_support c.conscientiousness##c.extraversion##i.framed
	// reject. 0.887	
	tess c.conscientiousness#c.extraversion#1.framed +
	
*Test-Hyp15: Openness will increase the effect of risk frame on opposition to Iran strike. 
	reg iranstrike_opp c.openness##i.framed
	// reject. 0.283
	tess c.openness#1.framed +
		
*Test-Hyp16: Openness will increase the effect of altruism frame on opposition to oil drilling. 
	reg oildrill_opp c.openness##i.framed
	// reject. 0.892
	tess c.openness#1.framed +
	
*Test-Hyp17: Openness will increase the effect of altruism frame on support for environmental regulations. 
	reg envreg_support c.openness##i.framed
	// reject. 0.293
	tess c.openness#1.framed +
	
	